# $Id: PKGBUILD 85755 2013-03-06 11:30:49Z jelle $
# Maintainer: Allan McRae <allan@archlinux.org>
# Contributor: William Rea <sillywilly@gmail.com>

pkgname=chromium-bsu
pkgver=0.9.15
pkgrel=4
pkgdesc="A fast paced top scrolling shooter"
arch=('i686' 'x86_64')
url="http://chromium-bsu.sourceforge.net/"
license=('custom:artistic')
depends=('freeglut' 'glu' 'sdl' 'openal' 'freealut' 'ftgl' 'fontconfig' 'sdl_image')
install=chromium-bsu.install
source=(http://downloads.sourceforge.net/project/$pkgname/Chromium%20B.S.U.%20source%20code/$pkgver/$pkgname-$pkgver.tar.gz)
md5sums=('d57137a1ac2292363b35192724c7e3be')

build(){
  cd $srcdir/$pkgname-$pkgver
  ./configure --prefix=/usr
  make
}

package() {
  cd $srcdir/$pkgname-$pkgver
  make DESTDIR=$pkgdir install

  install -Dm644 COPYING $pkgdir/usr/share/licenses/chromium-bsu/LICENSE
}

